



NAME: \_\_\_\_\_

## 1 Objective: MACROS and FILE I/O

Macros are very useful if you want to make your SAS programs more flexible and allow them to be used in different situations without having to rewrite entire programs. Macro variables also provide a useful method for passing information from one DATA step to another. Macros can be entire SAS programs or just pieces of SAS code. You start a macro with a %MACRO statement and end it with a %MEND (macro end) statement.

## 2 Key Words

%macro, infile, missing data.

## 3 MACRO

Try the following macro. Note especially the way array bounds are passed in line 2 of data step.

SAS code 3.1: Macro example

```
%MACRO ex1(nr, lc, hc, trig, rseed);  
    DATA ex_&trig;  
        array y[&lc:&hc] y&lc - y&hc;    /* <<< NOTE >>> */  
                                           /* y[*] gives out of bounds error when lc==0 */  
                                           /* Errors when lc < 0. Need 0<= lc <= hc */  
  
        do r=0 to &nr;  
            do c=&lc to &hc;  
                y[c] = &trig(r + c / 10);  
                mn = mean(of y[*]);  
            end;  
            z = rannor(&rseed);  
            output;  
        end;  
        keep r y&lc - y&hc mn z;  
    run;  
  
    title "ex_&trig(r+c/10), r=0:&nr, c=&lc:&hc";  
  
    PROC print data = ex_&trig;  
    run;  
  
    PROC means data = ex_&trig n mean std;  
        var y&lc - y&hc mn z;
```

```
run;

PROC sgplot data = ex_&trig;
    series X=r Y=y&lc;
run;
%mend ex1;

%ex1(10, 0, 4, sin, 1234)
%ex1(10, 2, 5, cos, 123)
%ex1(5, 5,13, tan,12)
```

- [1] Try using &lc < 0: what happens?
- [2] TURN IN: Write a short paragraph about what above macro does.

## 4 FILE I/O

**Given:** Save the file on Blackboard called `hus_wif.dat`, an ASCII file containing five columns: the age [yrs] and height [mm] of husbands and wives, and age of husband at marriage.

PROBLEM: The data set has some missing values, which appear as ‘ \* ’. SAS needs numerical missing values denoted as ‘ . ’ (a period).

SOLUTION 1: Put data set into EXCEL or word processor that can “find / replace”.

SOLUTION 2: Besides that, the data are tab delimited, and SAS demands only spaces, unless EXPANDTABS is used.

- [1] Process the data as in solution 1, and save it as `hus_wif.dat`.
- [2] Try the next SAS code on the adjusted text.

SAS code 4.1: Reading data from other files

```
DATA hs_wf;
    infile 'C:\....\hus_wif.dat' EXPANDTABS;
                                /* husband and wife's present age and ht, and h's age at marriage */
    input ha hh wa wh ham;
    ym = ha - ham;              /* yrs married */
    wam = wa - ym;
run;

PROC print data = hs_wf;
    title "hs_wf data";
run;
```

- [3] Notice how missing values propagate. Notice how we used data step to add other columns to data as it was being read in.
- [4] Read CS about `infile` to see more options of reading different types of files.